NAG Fortran Library Routine Document

F08KQF (ZGELSD)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08KQF (ZGELSD) computes the minimum norm solution to a real linear least-squares problem

$$\min_x \|b - Ax\|_2.$$

2 Specification

```
      SUBROUTINE F08KQF (M, N, NRHS, A, LDA, B, LDB, S, RCOND, RANK, WORK,
     1                   LWORK, RWORK, IWORK, INFO)
      INTEGER            M, N, NRHS, LDA, LDB, RANK, LWORK, IWORK(*), INFO
      double precision   S(*), RCOND, RWORK(*)
      complex*16         A(LDA,*), B(LDB,*), WORK(*)
```

The routine may be called by its LAPACK name ***zgelsd***.

3 Description

F08KQF (ZGELSD) uses the singular value decomposition (SVD) of A, where A is an m by n matrix which may be rank-deficient.

Several right-hand side vectors b and solution vectors x can be handled in a single call; they are stored as the columns of the m by r right-hand side matrix B and the n by r solution matrix X.

The problem is solved in three steps:

1. reduce the coefficient matrix A to bidiagonal form with Householder transformations, reducing the original problem into a 'bidiagonal least-squares problem' (BLS);
2. solve the BLS using a divide-and-conquer approach;
3. apply back all the Householder transformations to solve the original least-squares problem.

The effective rank of A is determined by treating as zero those singular values which are less than RCOND times the largest singular value.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia URL: http://www.netlib.org/lapack/lug

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

1: M – INTEGER *Input*

On entry: m, the number of rows of the matrix A.

Constraint: $M \geq 0$.

2: N – INTEGER *Input*

On entry: n, the number of columns of the matrix A.

Constraint: $N \geq 0$.

3: NRHS – INTEGER *Input*

On entry: r, the number of right-hand sides, i.e., the number of columns of the matrices B and X.

Constraint: $NRHS \geq 0$.

4: A(LDA,$*$) – ***complex*16*** array *Input/Output*

Note: the second dimension of the array A must be at least $\max(1, N)$.

On entry: the m by n matrix A.

On exit: the contents of A are destroyed.

5: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F08KQF (ZGELSD) is called.

Constraint: $LDA \geq \max(1, M)$.

6: B(LDB,$*$) – ***complex*16*** array *Input/Output*

Note: the second dimension of the array B must be at least $\max(1, NRHS)$.

On entry: the m by r right-hand side matrix B.

On exit: is overwritten by the n by r solution matrix X. If $m \geq n$ and $RANK = n$, the residual sum of squares for the solution in the ith column is given by the sum of squares of the modulus of elements $n+1, \ldots, m$ in that column.

7: LDB – INTEGER *Input*

On entry: the first dimension of the array B as declared in the (sub)program from which F08KQF (ZGELSD) is called.

Constraint: $LDB \geq \max(1, M, N)$.

8: S($*$) – ***double precision*** array *Output*

Note: the dimension of the array S must be at least $\max(1, \min(M, N))$.

On exit: the singular values of A in decreasing order.

9: RCOND – ***double precision*** *Input*

On entry: used to determine the effective rank of A. Singular values $S(i) \leq RCOND \times S(1)$ are treated as zero. If $RCOND < 0$, ***machine precision*** is used instead.

10: RANK – INTEGER *Output*

On exit: the effective rank of A, i.e., the number of singular values which are greater than $RCOND \times S(1)$.

11: WORK($*$) – ***complex*16*** array *Workspace*

Note: the dimension of the array WORK must be at least $\max(1, LWORK)$.

On exit: if $INFO = 0$, the real part of WORK(1) contains the minimum value of LWORK required for optimal performance.

12: LWORK – INTEGER *Input*

On entry: the dimension of the array WORK as declared in the (sub)program from which F08KQF (ZGELSD) is called.

The exact minimum amount of workspace needed depends on M, N and NRHS. As long as LWORK is at least

$$\max(1, \mathrm{M}+\mathrm{N}+r, 2r + r\times \mathrm{NRHS}),$$

where $r = \min(\mathrm{M}, \mathrm{N})$, the code will execute correctly.

If LWORK $= -1$, a workspace query is assumed; the routine only calculates the optimal size of the WORK array and the minimum size of the IWORK array, and returns these values as the first entries of the WORK and IWORK arrays, and no error message related to LWORK is issued.

Suggested value: for optimal performance, LWORK should generally be larger than the required minimum. Consider increasing LWORK by at least $nb \times \min(\mathrm{M}, \mathrm{N})$, where nb is the optimal ***block size***.

Constraint: LWORK must be at least $\max(1, \mathrm{M}+\mathrm{N}+r, 2r + r\times \mathrm{NRHS})$ or LWORK $= -1$.

13: RWORK($*$) – ***double precision*** array *Workspace*

Note: the dimension of the array RWORK must be at least $\max(1, lrwork)$, where $lrwork$ is at least

$$10\times \mathrm{N} + 2\times \mathrm{N}\times smlsiz + 8\times \mathrm{N}\times nlvl + 3\times smlsiz \times \mathrm{NRHS} + (smlsiz+1)^2, \quad \text{if } \mathrm{M}\geq \mathrm{N}$$

or

$$10\times \mathrm{M} + 2\times \mathrm{M}\times smlsiz + 8\times \mathrm{M}\times nlvl + 3\times smlsiz \times \mathrm{NRHS} + (smlsiz+1)^2, \quad \text{if } \mathrm{M} < \mathrm{N}$$

where $smlsiz$ is equal to the maximum size of the subproblems at the bottom of the computation tree (usually about 25), and $nlvl = \max(0, \mathrm{int}(\log_2(\min(\mathrm{M},\mathrm{N})/(smlsiz+1))) + 1)$, the code will execute correctly.

On exit: if INFO $= 0$, RWORK(1) contains the required minimal size of $lrwork$.

14: IWORK($*$) – INTEGER array *Workspace*

Note: the dimension of the array IWORK must be at least $\max(1, liwork)$, where $liwork$ is at least $\max(1, 3\times \min(\mathrm{M},\mathrm{N})\times nlvl + 11\times \min(\mathrm{M},\mathrm{N}))$.

On exit: if INFO $= 0$, IWORK(1) returns the minimum $liwork$.

15: INFO – INTEGER *Output*

On exit: INFO $= 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, the ith parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

The algorithm for computing the SVD failed to converge; if INFO $= i$, i off-diagonal elements of an intermediate bidiagonal form did not converge to zero.

7 Accuracy

See Section 4.5 of Anderson *et al.* (1999) for details.

8 Further Comments

The real analogue of this routine is F08KCF (DGELSD).

9 Example

This example solves the linear least-squares problem

$$\min_x \|b - Ax\|_2$$

for the solution, x, of minimum norm, where

$$A = \begin{pmatrix} 0.47 - 0.34i & -0.32 - 0.23i & 0.35 - 0.60i & 0.89 + 0.71i & -0.19 + 0.06i \\ -0.40 + 0.54i & -0.05 + 0.20i & -0.52 - 0.34i & -0.45 - 0.45i & 0.11 - 0.85i \\ 0.60 + 0.01i & -0.26 - 0.44i & 0.87 - 0.11i & -0.02 - 0.57i & 1.44 + 0.80i \\ 0.80 - 1.02i & -0.43 + 0.17i & -0.34 - 0.09i & 1.14 - 0.78i & 0.07 + 1.14i \end{pmatrix}$$

and

$$b = \begin{pmatrix} 2.15 - 0.20i \\ -2.24 + 1.82i \\ 4.45 - 4.28i \\ 5.70 - 6.25i \end{pmatrix}.$$

A tolerance of 0.01 is used to determine the effective rank of A.

Note that the block size (NB) of 64 assumed in this example is not realistic for such a small problem, but should be suitable for large problems.

9.1 Program Text

```
*     F08KQF Example Program Text
*     Mark 21 Release. NAG Copyright 2004.
*     .. Parameters ..
*    $                 LRWORK=1
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          MMAX, NB, NLVL, NMAX, SMLSIZ
      PARAMETER        (MMAX=8,NB=64,NLVL=10,NMAX=16,SMLSIZ=25)
      INTEGER          LDA, LIWORK, LRWORK, LWORK
      PARAMETER        (LDA=MMAX,LIWORK=3*MMAX*NLVL+11*MMAX,
     +                 LRWORK=10*MMAX+2*MMAX*SMLSIZ+8*MMAX*NLVL+3*
     +                 SMLSIZ+(SMLSIZ+1)**2,LWORK=2*MMAX+NB*(MMAX+NMAX))
*     .. Local Scalars ..
      DOUBLE PRECISION RCOND
      INTEGER          I, INFO, J, M, N, RANK
*     .. Local Arrays ..
      COMPLEX *16      A(LDA,NMAX), B(NMAX), WORK(LWORK)
      DOUBLE PRECISION RWORK(LRWORK), S(MMAX)
      INTEGER          IWORK(LIWORK)
*     .. External Subroutines ..
      EXTERNAL         ZGELSD
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F08KQF Example Program Results'
      WRITE (NOUT,*)
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) M, N
      IF (M.LE.MMAX .AND. N.LE.NMAX .AND. M.LE.N) THEN
*
*        Read A and B from data file
*
         READ (NIN,*) ((A(I,J),J=1,N),I=1,M)
         READ (NIN,*) (B(I),I=1,M)
*
*        Choose RCOND to reflect the relative accuracy of the input
*        data
```

```
*
         RCOND = 0.01D0
*
*        Solve the least squares problem min( norm2(b - Ax) ) for the
*        x of minimum norm.
*
         CALL ZGELSD(M,N,1,A,LDA,B,N,S,RCOND,RANK,WORK,LWORK,RWORK,
     +               IWORK,INFO)
*
         IF (INFO.EQ.0) THEN
*
*           Print solution
*
            WRITE (NOUT,*) 'Least squares solution'
            WRITE (NOUT,99999) (B(I),I=1,N)
*
*           Print the effective rank of A
*
            WRITE (NOUT,*)
            WRITE (NOUT,*) 'Tolerance used to estimate the rank of A'
            WRITE (NOUT,99998) RCOND
            WRITE (NOUT,*) 'Estimated rank of A'
            WRITE (NOUT,99997) RANK
*
*           Print singular values of A
*
            WRITE (NOUT,*)
            WRITE (NOUT,*) 'Singular values of A'
            WRITE (NOUT,99996) (S(I),I=1,M)
         ELSE IF (INFO.GT.0) THEN
            WRITE (NOUT,*) 'The SVD algorithm failed to converge'
         END IF
      ELSE
         WRITE (NOUT,*) 'MMAX and/or NMAX too small, and/or M.GT.N'
      END IF
      STOP
*
99999 FORMAT (4(' (',F7.4,',',F7.4,')',:))
99998 FORMAT (3X,1P,E11.2)
99997 FORMAT (1X,I6)
99996 FORMAT (1X,7F11.4)
      END
```

9.2 Program Data

```
F08KQF Example Program Data

   4           5                                          :Values of M and N

 ( 0.47,-0.34) (-0.32,-0.23) ( 0.35,-0.60) ( 0.89, 0.71) (-0.19, 0.06)
 (-0.40, 0.54) (-0.05, 0.20) (-0.52,-0.34) (-0.45,-0.45) ( 0.11,-0.85)
 ( 0.60, 0.01) (-0.26,-0.44) ( 0.87,-0.11) (-0.02,-0.57) ( 1.44, 0.80)
 ( 0.80,-1.02) (-0.43, 0.17) (-0.34,-0.09) ( 1.14,-0.78) ( 0.07, 1.14) :End of A

 ( 2.15,-0.20)
 (-2.24, 1.82)
 ( 4.45,-4.28)
 ( 5.70,-6.25)                                              :End of vector b
```

9.3 Program Results

```
 F08KQF Example Program Results

 Least squares solution
 ( 3.9747,-1.8377) (-0.9186, 0.8253) (-0.3105, 0.1477) ( 1.0050, 0.8626)
 (-0.2256,-1.9425)

 Tolerance used to estimate the rank of A
      1.00E-02
 Estimated rank of A
      3

 Singular values of A
      2.9979     1.9983     1.0044     0.0064
```